
NAG Library Routine Document

C05QCF

Note: before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold italicised terms and
other implementation-dependent details.

1 Purpose

C05QCF is a comprehensive routine that finds a solution of a system of nonlinear equations by a
modification of the Powell hybrid method.

2 Specification

SUBROUTINE C05QCF (FCN, N, X, FVEC, XTOL, MAXFEV, ML, MU, EPSFCN, MODE,
DIAG, FACTOR, NPRINT, NFEV, FJAC, R, QTF, IUSER, RUSER,
IFAIL)

&
&

INTEGER N, MAXFEV, ML, MU, MODE, NPRINT, NFEV, IUSER(*), IFAIL

REAL (KIND=nag_wp) X(N), FVEC(N), XTOL, EPSFCN, DIAG(N), FACTOR,
FJAC(N,N), R(N*(N+1)/2), QTF(N), RUSER(*)

&

EXTERNAL FCN

3 Description

The system of equations is defined as:

fi x1; x2; . . . ; xnð Þ ¼ 0, i ¼ 1; 2; . . . ; n.

C05QCF is based on the MINPACK routine HYBRD (see Moré et al. (1980)). It chooses the correction at
each step as a convex combination of the Newton and scaled gradient directions. The Jacobian is updated
by the rank-1 method of Broyden. At the starting point, the Jacobian is approximated by forward
differences, but these are not used again until the rank-1 method fails to produce satisfactory progress. For
more details see Powell (1970).

4 References

Moré J J, Garbow B S and Hillstrom K E (1980) User guide for MINPACK-1 Technical Report ANL-80-74
Argonne National Laboratory

Powell M J D (1970) A hybrid method for nonlinear algebraic equations Numerical Methods for Nonlinear
Algebraic Equations (ed P Rabinowitz) Gordon and Breach

5 Parameters

1: FCN – SUBROUTINE, supplied by the user. External Procedure

FCN must return the values of the functions fi at a point x, unless IFLAG ¼ 0 on entry to C05QCF.

The specification of FCN is:

SUBROUTINE FCN (N, X, FVEC, IUSER, RUSER, IFLAG)

INTEGER N, IUSER(*), IFLAG

REAL (KIND=nag_wp) X(N), FVEC(N), RUSER(*)

1: N – INTEGER Input

On entry: n, the number of equations.

2: XðNÞ – REAL (KIND=nag_wp) array Input

On entry: the components of the point x at which the functions must be evaluated.

3: FVECðNÞ – REAL (KIND=nag_wp) array Input/Output

On entry: if IFLAG ¼ 0, FVEC contains the function values fi xð Þ and must not be
changed.

On exit: if IFLAG > 0 on entry, FVEC must contain the function values fi xð Þ (unless
IFLAG is set to a negative value by FCN).

4: IUSERð�Þ – INTEGER array User Workspace
5: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

FCN is called with the parameters IUSER and RUSER as supplied to C05QCF. You are
free to use the arrays IUSER and RUSER to supply information to FCN as an alternative
to using COMMON global variables.

6: IFLAG – INTEGER Input/Output

On entry: IFLAG � 0.

IFLAG ¼ 0

X and FVEC are available for printing (see NPRINT).

IFLAG > 0

FVEC must be updated.

On exit: in general, IFLAG should not be reset by FCN. If, however, you wish to
terminate execution (perhaps because some illegal point X has been reached), then IFLAG
should be set to a negative integer.

FCN must either be a module subprogram USEd by, or declared as EXTERNAL in, the
(sub)program from which C05QCF is called. Parameters denoted as Input must not be changed by
this procedure.

2: N – INTEGER Input

On entry: n, the number of equations.

Constraint: N > 0.

3: XðNÞ – REAL (KIND=nag_wp) array Input/Output

On entry: an initial guess at the solution vector.

On exit: the final estimate of the solution vector.

4: FVECðNÞ – REAL (KIND=nag_wp) array Output

On exit: the function values at the final point returned in X.

5: XTOL – REAL (KIND=nag_wp) Input

On entry: the accuracy in X to which the solution is required.

Suggested value:
ffiffi

�
p

, where � is the machine precision returned by X02AJF.

Constraint: XTOL � 0:0.

6: MAXFEV – INTEGER Input

On entry: the maximum number of calls to FCN with IFLAG 6¼ 0. C05QCF will exit with
IFAIL ¼ 2, if, at the end of an iteration, the number of calls to FCN exceeds MAXFEV.

Suggested value: MAXFEV ¼ 200� Nþ 1ð Þ.
Constraint: MAXFEV > 0.

7: ML – INTEGER Input

On entry: the number of subdiagonals within the band of the Jacobian matrix. (If the Jacobian is
not banded, or you are unsure, set ML ¼ N� 1.)

Constraint: ML � 0.

8: MU – INTEGER Input

On entry: the number of superdiagonals within the band of the Jacobian matrix. (If the Jacobian is
not banded, or you are unsure, set MU ¼ N� 1.)

Constraint: MU � 0.

9: EPSFCN – REAL (KIND=nag_wp) Input

On entry: a rough estimate of the largest relative error in the functions. It is used in determining a
suitable step for a forward difference approximation to the Jacobian. If EPSFCN is less than
machine precision (returned by X02AJF) then machine precision is used. Consequently a value of
0:0 will often be suitable.

Suggested value: EPSFCN ¼ 0:0.

10: MODE – INTEGER Input

On entry: indicates whether or not you have provided scaling factors in DIAG.

If MODE ¼ 2 the scaling must have been specified in DIAG.

Otherwise, if MODE ¼ 1, the variables will be scaled internally.

Constraint: MODE ¼ 1 or 2.

11: DIAGðNÞ – REAL (KIND=nag_wp) array Input/Output

On entry: if MODE ¼ 2, DIAG must contain multiplicative scale factors for the variables.

If MODE ¼ 1, DIAG need not be set.

Constraint: if MODE ¼ 2, DIAGðiÞ > 0:0, for i ¼ 1; 2; . . . ; n.

On exit: the scale factors actually used (computed internally if MODE ¼ 1).

12: FACTOR – REAL (KIND=nag_wp) Input

On entry: a quantity to be used in determining the initial step bound. In most cases, FACTOR
should lie between 0:1 and 100:0. (The step bound is FACTOR� DIAG� Xk k2 if this is nonzero;
otherwise the bound is FACTOR.)

Suggested value: FACTOR ¼ 100:0.

Constraint: FACTOR > 0:0.

13: NPRINT – INTEGER Input

On entry: indicates whether (and how often) special calls to FCN, with IFLAG set to 0, are to be
made for printing purposes.

NPRINT � 0
No calls are made.

NPRINT > 0
FCN is called at the beginning of the first iteration, every NPRINT iterations thereafter and
immediately before the return from C05QCF.

14: NFEV – INTEGER Output

On exit: the number of calls made to FCN with IFLAG > 0.

15: FJACðN,NÞ – REAL (KIND=nag_wp) array Output

On exit: the orthogonal matrix Q produced by the QR factorisation of the final approximate
Jacobian.

16: RðN� Nþ 1ð Þ=2Þ – REAL (KIND=nag_wp) array Output

On exit: the upper triangular matrix R produced by the QR factorization of the final approximate
Jacobian, stored row-wise.

17: QTFðNÞ – REAL (KIND=nag_wp) array Output

On exit: the vector QTf.

18: IUSERð�Þ – INTEGER array User Workspace
19: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

IUSER and RUSER are not used by C05QCF, but are passed directly to FCN and may be used to
pass information to this routine as an alternative to using COMMON global variables.

20: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this parameter you should
refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit (as
defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 2

There have been at least MAXFEV evaluations of FCN. Consider restarting the calculation from the
final point held in X.

IFAIL ¼ 3

No further improvement in the approximate solution X is possible; XTOL is too small.

IFAIL ¼ 4

The iteration is not making good progress, as measured by the improvement from the last five
Jacobian evaluations.

IFAIL ¼ 5

The iteration is not making good progress, as measured by the improvement from the last ten
iterations.

IFAIL ¼ 6

You have set IFLAG negative in FCN.

IFAIL ¼ 11

On entry, N � 0.

IFAIL ¼ 12

On entry, XTOL < 0:0.

IFAIL ¼ 13

On entry, MODE 6¼ 1 or 2.

IFAIL ¼ 14

On entry, FACTOR � 0:0.

IFAIL ¼ 15

On entry, MODE ¼ 2 and DIAGðiÞ � 0:0 for some i, for i ¼ 1; 2; . . . ;N.

IFAIL ¼ 16

On entry, ML < 0.

IFAIL ¼ 17

On entry, MU < 0.

IFAIL ¼ 18

On entry, MAXFEV � 0.

IFAIL ¼ �999

Internal memory allocation failed.

A value of IFAIL ¼ 4 or 5 may indicate that the system does not have a zero, or that the solution is very
close to the origin (see Section 7). Otherwise, rerunning C05QCF from a different starting point may
avoid the region of difficulty.

7 Accuracy

If x̂ is the true solution and D denotes the diagonal matrix whose entries are defined by the array DIAG,
then C05QCF tries to ensure that

D x� x̂ð Þk k2 � XTOL� Dx̂k k2.

If this condition is satisfied with XTOL ¼ 10�k, then the larger components of Dx have k significant
decimal digits. There is a danger that the smaller components of Dx may have large relative errors, but
the fast rate of convergence of C05QCF usually obviates this possibility.

If XTOL is less than machine precision and the above test is satisfied with the machine precision in place
of XTOL, then the routine exits with IFAIL ¼ 3.

Note: this convergence test is based purely on relative error, and may not indicate convergence if the
solution is very close to the origin.

The convergence test assumes that the functions are reasonably well behaved. If this condition is not
satisfied, then C05QCF may incorrectly indicate convergence. The validity of the answer can be checked,
for example, by rerunning C05QCF with a lower value for XTOL.

8 Further Comments

Local workspace arrays of fixed lengths are allocated internally by C05QCF. The total size of these arrays
amounts to 4� n real elements.

The time required by C05QCF to solve a given problem depends on n, the behaviour of the functions, the
accuracy requested and the starting point. The number of arithmetic operations executed by C05QCF to

process each evaluation of the functions is approximately 11:5� n2. The timing of C05QCF is strongly
influenced by the time spent evaluating the functions.

Ideally the problem should be scaled so that, at the solution, the function values are of comparable
magnitude.

The number of function evaluations required to evaluate the Jacobian may be reduced if you can specify
ML and MU accurately.

9 Example

This example determines the values x1; . . . ; x9 which satisfy the tridiagonal equations:

3� 2x1ð Þx1 � 2x2 ¼ �1,
�xi�1 þ 3� 2xið Þxi � 2xiþ1 ¼ �1, i ¼ 2; 3; . . . ; 8

�x8 þ 3� 2x9ð Þx9 ¼ �1.

9.1 Program Text

! C05QCF Example Program Text
! Mark 24 Release. NAG Copyright 2012.

Module c05qcfe_mod

! C05QCF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Real (Kind=nag_wp), Parameter :: epsfcn = 0.0_nag_wp
Real (Kind=nag_wp), Parameter :: factor = 100.0_nag_wp
Integer, Parameter :: maxfev = 2000, ml = 1, mode = 2, &

mu = 1, n = 9, nout = 6, &
nprint = 0

Contains
Subroutine fcn(n,x,fvec,iuser,ruser,iflag)

! .. Scalar Arguments ..
Integer, Intent (Inout) :: iflag
Integer, Intent (In) :: n

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: fvec(n), ruser(*)
Real (Kind=nag_wp), Intent (In) :: x(n)
Integer, Intent (Inout) :: iuser(*)

! .. Executable Statements ..
If (iflag==0) Then

! Insert print statements here when NPRINT is positive.

Continue
Else

fvec(1:n) = (3.0_nag_wp-2.0_nag_wp*x(1:n))*x(1:n) + 1.0_nag_wp
fvec(2:n) = fvec(2:n) - x(1:(n-1))
fvec(1:(n-1)) = fvec(1:(n-1)) - 2.0_nag_wp*x(2:n)

End If

Return

End Subroutine fcn
End Module c05qcfe_mod
Program c05qcfe

! C05QCF Example Main Program

! .. Use Statements ..
Use nag_library, Only: c05qcf, dnrm2, nag_wp, x02ajf
Use c05qcfe_mod, Only: epsfcn, factor, fcn, maxfev, ml, mode, mu, n, &

nout, nprint
! .. Implicit None Statement ..

Implicit None
! .. Local Scalars ..

Real (Kind=nag_wp) :: fnorm, xtol
Integer :: ifail, j, nfev

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: diag(:), fjac(:,:), fvec(:), &

qtf(:), r(:), x(:)
Real (Kind=nag_wp) :: ruser(1)
Integer :: iuser(1)

! .. Intrinsic Procedures ..
Intrinsic :: sqrt

! .. Executable Statements ..
Write (nout,*) ’C05QCF Example Program Results’

Allocate (diag(n),fjac(n,n),fvec(n),qtf(n),r(n*(n+1)/2),x(n))

! The following starting values provide a rough solution.

x(1:n) = -1.0_nag_wp
xtol = sqrt(x02ajf())
diag(1:n) = 1.0_nag_wp

ifail = -1
Call c05qcf(fcn,n,x,fvec,xtol,maxfev,ml,mu,epsfcn,mode,diag,factor, &

nprint,nfev,fjac,r,qtf,iuser,ruser,ifail)

Select Case (ifail)
Case (0)

! The NAG name equivalent of dnrm2 is f06ejf
fnorm = dnrm2(n,fvec,1)
Write (nout,*)
Write (nout,99999) ’Final 2-norm of the residuals =’, fnorm
Write (nout,*)
Write (nout,99998) ’Number of function evaluations =’, nfev
Write (nout,*)
Write (nout,*) ’Final approximate solution’
Write (nout,*)
Write (nout,99997)(x(j),j=1,n)

Case (2:5)
Write (nout,*)
Write (nout,*) ’Approximate solution’
Write (nout,*)
Write (nout,99997)(x(j),j=1,n)

End Select

99999 Format (1X,A,E12.4)
99998 Format (1X,A,I10)
99997 Format (1X,3F12.4)

End Program c05qcfe

9.2 Program Data

None.

9.3 Program Results

C05QCF Example Program Results

Final 2-norm of the residuals = 0.1193E-07

Number of function evaluations = 14

Final approximate solution

-0.5707 -0.6816 -0.7017
-0.7042 -0.7014 -0.6919
-0.6658 -0.5960 -0.4164

___________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________

